Помогите убрать задвоение строк в запросе. Есть Объединенный запрос, не могу убрать задвоение. Подскажите как привильней настроить группировку и итоги.

```1c
    ВЫБРАТЬ РАЗЛИЧНЫЕ
    ТоварыОрганизацийОстатки.Склад КАК Склад,
    ТоварыОрганизацийОстатки.Номенклатура КАК Номенклатура,
    0 КАК КоличествоОстаток2,
    СУММА(ТоварыОрганизацийОстатки.КоличествоОстаток) КАК КоличествоОстаток1
    ИЗ
    РегистрНакопления.ТоварыОрганизаций.Остатки КАК ТоварыОрганизацийОстатки
    ГДЕ
    ТоварыОрганизацийОстатки.Организация = &Организация
    И ТоварыОрганизацийОстатки.Склад = &Склад
    И ТоварыОрганизацийОстатки.Номенклатура В ИЕРАРХИИ(&Номенклатура)

    СГРУППИРОВАТЬ ПО
    ТоварыОрганизацийОстатки.Склад,
    ТоварыОрганизацийОстатки.Номенклатура

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ТоварыОрганизацийОстатки.Склад,
    ТоварыОрганизацийОстатки.Номенклатура,
    СУММА(ТоварыОрганизацийОстатки.КоличествоОстаток),
    0
    ИЗ
    РегистрНакопления.ТоварыОрганизаций.Остатки КАК ТоварыОрганизацийОстатки
    ГДЕ
    ТоварыОрганизацийОстатки.Организация = &Организация2
    И ТоварыОрганизацийОстатки.Склад = &Склад
    И ТоварыОрганизацийОстатки.Номенклатура В ИЕРАРХИИ(&Номенклатура)

    СГРУППИРОВАТЬ ПО
    ТоварыОрганизацийОстатки.Склад,
    ТоварыОрганизацийОстатки.Номенклатура
    ИТОГИ
    СУММА(КоличествоОстаток2),
    СУММА(КоличествоОстаток1)
    ПО
    Склад,
    Номенклатура
```
---
Вот запрос, который тебе нужен:

```1c
    ВЫБРАТЬ
    ТоварыОрганизацийОстатки.Склад КАК Склад,
    ТоварыОрганизацийОстатки.Номенклатура КАК Номенклатура,
    0 КАК КоличествоОстаток2,
    ТоварыОрганизацийОстатки.КоличествоОстаток КАК КоличествоОстаток1
    ПОМЕСТИТЬ ВТ
    ИЗ
    РегистрНакопления.ТоварыОрганизаций.Остатки(
    ,
    Организация = &Организация
    И Номенклатура В ИЕРАРХИИ (&Номенклатура)) КАК ТоварыОрганизацийОстатки

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ТоварыОрганизацийОстатки.Склад,
    ТоварыОрганизацийОстатки.Номенклатура,
    ТоварыОрганизацийОстатки.КоличествоОстаток,
    0
    ИЗ
    РегистрНакопления.ТоварыОрганизаций.Остатки(
    ,
    Организация = &Организация2
    И Номенклатура В ИЕРАРХИИ (&Номенклатура)) КАК ТоварыОрганизацийОстатки
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ВТ.Номенклатура КАК Номенклатура,
    ВТ.Склад КАК Склад,
    СУММА(ВТ.КоличествоОстаток2) КАК КоличествоОстаток2,
    СУММА(ВТ.КоличествоОстаток1) КАК КоличествоОстаток1
    ИЗ
    ВТ КАК ВТ

    СГРУППИРОВАТЬ ПО
    ВТ.Склад,
    ВТ.Номенклатура
```
---
итоги убрать. ИТОГИ
СУММА(КоличествоОстаток2),
СУММА(КоличествоОстаток1)
ПО
Склад,
Номенклатура

Это не задвоение, а итоги по группировкам.
Достаточно установить вывод в дерево и будет сразу видно.
---
убирая итоги, задвоение убирается, но в результате пишет примерно вот так
сироп карамель 14
сироп карамель -1
а мне нужно что бы было как то так
сироп карамель 14 -1
---
а на этот вопрос уже ответили в (2)
---
поставил дерево. выглядит прилично, у меня теперь такой вопрос если я буду выводить результат в табличную части внешнее обработки то в строку будет передаваться раскрытая ветка или нет например
1 строка сироп 14 -1
или будет так
1 строка сироп 14 -1
2 строка сироп 14
3 строка сироп -1
---
Объединение поместить во вложенный запрос или временную таблицу, получившуюся итоговую таблицу сгруппировать и вывести итоги
---
Вот запрос, который тебе нужен:

```1c
    ВЫБРАТЬ
    ТоварыОрганизацийОстатки.Склад КАК Склад,
    ТоварыОрганизацийОстатки.Номенклатура КАК Номенклатура,
    0 КАК КоличествоОстаток2,
    ТоварыОрганизацийОстатки.КоличествоОстаток КАК КоличествоОстаток1
    ПОМЕСТИТЬ ВТ
    ИЗ
    РегистрНакопления.ТоварыОрганизаций.Остатки(
    ,
    Организация = &Организация
    И Номенклатура В ИЕРАРХИИ (&Номенклатура)) КАК ТоварыОрганизацийОстатки

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ТоварыОрганизацийОстатки.Склад,
    ТоварыОрганизацийОстатки.Номенклатура,
    ТоварыОрганизацийОстатки.КоличествоОстаток,
    0
    ИЗ
    РегистрНакопления.ТоварыОрганизаций.Остатки(
    ,
    Организация = &Организация2
    И Номенклатура В ИЕРАРХИИ (&Номенклатура)) КАК ТоварыОрганизацийОстатки
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ВТ.Номенклатура КАК Номенклатура,
    ВТ.Склад КАК Склад,
    СУММА(ВТ.КоличествоОстаток2) КАК КоличествоОстаток2,
    СУММА(ВТ.КоличествоОстаток1) КАК КоличествоОстаток1
    ИЗ
    ВТ КАК ВТ

    СГРУППИРОВАТЬ ПО
    ВТ.Склад,
    ВТ.Номенклатура
```
---
спасибо огромное! только учусь, еще со не работал со временными таблицами, очень помогли!
---
Обрати внимание, что параметры запроса из секции ГДЕ, я перенес в параметры виртуальной таблицы (т. е. в скобочки):

```1c
    РегистрНакопления.ТоварыОрганизаций.Остатки(, Организация = &Организация И Номенклатура В ИЕРАРХИИ (&Номенклатура))
```

ибо твой вариант совсем плох был